AN250

Auto-calibration of the PIC12F6XX Internal RC Oscillator to ± 1%

Author: *Mike Rylee*
Microchip Technology Inc.

INTRODUCTION

This application note describes a technique used to calibrate the internal oscillator on the PIC12F6XX devices. This technique allows the internal oscillator to be calibrated within ± 1%. See the test circuit shown in Figure 1.

This application note is useful in the following applications:

1. High volume production environment.
2. Battery applications using on-board calibration to recalibrate the internal RC oscillator as the battery voltage drops.
3. Applications that are exposed to a varying voltage and temperature range could have intelligent on-board recalibration.

FIGURE 1: TEST CIRCUIT



THEORY OF OPERATION

Equipment Used

- HP54645D Oscilloscope with digital probes
- AGILENT 33120A Function Generator
- HPE3620A Power Supply
- Digital multimeter with frequency counter

Definitions

- **Tolerance** - A firmware specified value indicating the allowed deviation in the measured period of a 5 kHz square wave output from a fixed frequency source.
- **Calibration Counter** - A firmware specified value indicating how many times to perform a calibration.

Testing

To test the PIC12F6XX device, the firmware first determines if a calibration was performed. If a calibration was performed, the firmware loads the new value from EEPROM into the OSCCAL register. Otherwise, the original factory value is loaded into the OSCCAL register. If the device is calibrated, a 5 kHz square wave is generated by the PIC12F6XX and output on an I/O pin. By hooking up an oscilloscope or frequency counter to this pin, one can determine how precise the internal oscillator is calibrated by measuring the deviation of the frequency, or period of the square wave being output when compared to 5 kHz. See Figure 2 and Figure 3.

FIGURE 2: MAIN ROUTINE



FIGURE 3: TEST ROUTINE



Calibration

To calibrate the PIC12F6XX device, a 5 kHz 50% duty cycle square wave is injected into an I/O pin on the PIC12F6XX device. The firmware measures the period of the 5 kHz square wave and checks to see if it is within a specified tolerance. If the firmware determines that the PIC12F6XX internal oscillator is running faster than the specified tolerance, then it will decrement the value in the OSCCAL register by 1. If the firmware determines that the PIC12F6XX internal oscillator is running slower than the specified tolerance, then it will increment the value in the OSCCAL register by 1. The PIC12F6XX firmware will repeat the above procedure until the calibration counter reaches 0. After the firmware is finished calibrating the internal oscillator, the new calibration value is stored in EEPROM. See Figure 4 and Figure 5.

FIGURE 4: CALIBRATE ROUTINE



FIGURE 5: CHECKPERIOD() ROUTINE



CONCLUSION

This application note shows a method for auto-calibration of the PIC12F6XX internal RC oscillator. In particular, using a fixed frequency source allowed calibration of the PIC12F6XX internal RC oscillator to be within ± 1%.

APPENDIX A: PIC12F6XX AUTO-CALIBRATION & TEST PROGRAM

```
;*******************************************************************************************
;PIC12F6XX Autocalibration & Test Program
;
;Version:  1.0
;Date:     10/21/02
;Author:   Mike Rylee
;Description:This program calibrates the internal RC oscillator on the PIC12F6XX to the tolerance
;            specified in the #define TOLERANCE measurement parameter below. The oscillator is
;            calibrated using a 5kHz 50% square wave signal input on GP0 from a function generator.
;            The calibration is started by pushing the pushbutton connected to GP2.
;            After calibration a 5kHz test signal can be output on GP5 by the pushbutton connected
;            to GP1 to check the calibration.
;*******************************************************************************************

      list      p=12f675
      #include <p12f675.inc>

      __CONFIG   _CP_OFF & _CPD_OFF & _BODEN_OFF & _MCLRE_OFF & _WDT_OFF & _PWRTE_ON &
      INTRC_OSC_NOCLKOUT

;GENERAL PURPOSE REGISTERS
COUNTER EQU   0X20

;MEASUREMENT PARAMETERS
#define   REFERENCEPERIOD      .200      ;Reference Period In Microseconds
#define   TOLERANCE            .1        ;Tolerance Can Be Tweaked For Oscillator Accuracy
#define   CALIBRATIONTIME      .10       ;Number Of Times To Measure Signal During Calibration

;MAIN I/O
#define   INPUT0               GPIO,0    ;Input Reference Waveform Of 5KHz
#define   INPUT1               GPIO,1    ;Test Button
#define   INPUT2               GPIO,2    ;Calibrate Button
#define   OUTPUT               GPIO,5    ;Outputs 5KHz Calibrated Waveform

;EEPROM Definitions
#define   CALFLAGADR           0x7E      ;Value of 0xA5 => Calibration Has Not Been Performed
                                         ;Value of 0x5A => Calibration Has Been Performed
#define   CALVALADR            0X7F      ;Calibration Value

;DEBUG
#define   LED                  GPIO,4    ;Debug LED

;*******************************************************************************************
;Reset Vector
;*******************************************************************************************
          ORG                  0x000
          call                 0x3FF     ; Retrieve Factory Calibration Value
          BANKSEL              OSCCAL    ; BANK1
          movwf                OSCCAL    ; Load OSCCAL
```

```
         goto                INIT

;*****************************************************************************************
;Initialization
;*****************************************************************************************
INIT
         movlw               b'00001111'  ;GP0-Input, GP1-Input, GP2-Input, GP4-Output,GP5-Output
         movwf               TRISIO
         clrf                ANSEL
         clrf                VRCON        ;Turn Off VREF
         BANKSEL             GPIO         ;BANK 0

         movlw               .7           ;Turn Off comparator
         movwf               CMCON

         BANKSEL             OPTION_REG   ;BANK 1
         movlw               b'01001000'  ;Pull Ups Enabled, Rising Edge, Assigned to WDT,
                                          ;Prescaler is 1:1 WDT
         movwf               OPTION_REG
         bsf                 WPU,2        ;GP2 - Pullup Enabled
         bsf                 WPU,1        ;GP1 - Pullup Enabled
         BANKSEL             GPIO         ;BANK0
         clrf                GPIO

;*****************************************************************************************
;Main Program - This routine watches button INPUT1 & INPUT2
;             - Calls CALIBRATE() or Calls TEST()
;*****************************************************************************************
MAIN
         btfss               INPUT1       ;Check Test Button
         goto                ONE
         btfsc               INPUT2       ;Check Calibration Button
         goto                MAIN

TWO
         bsf                 LED
         call                CALIBRATE
         bcf                 LED
         goto                MAIN

ONE
         bsf                 LED
         call                TEST
         bcf                 LED
         goto                MAIN


;*****************************************************************************************
;Subroutines
;*****************************************************************************************
;TEST() -  This routine is used to test the OSCCAL value in the PIC12F6XX
;       -  Checks If Calibration Was Performed
;       -  Updates OSCCAL Value If Calibration Was Performed
;       -  Outputs A 5 kHZ 50% Square Wave On OUTPUT Until Pushbutton Is Released
;*****************************************************************************************
TEST
           movlw             CALFLAGADR
           call              EEREAD
           sublw             0x5A
           btfss             STATUS,Z     ;Was Calibration Flag Set?
           goto              STARTTEST    ;No Don't Change Osccal

           movlw             CALVALADR
           call              EEREAD       ;Yes Change The Osccal
           BANKSEL           OSCCAL       ;BANK1
```

```
            movwf           OSCCAL
            BANKSEL         GPIO        ;BANK0

STARTTEST                               ;The instructions below make a 5kHZ 50% Square Wave
                                        ;Output On GP1 if the device is calibrated
            bsf             OUTPUT      ;1 us
            movlw           .31         ;Delay 99 us
            call            DELAYUS     ;99+1 = 100 us
            bcf             OUTPUT      ;1us
            movlw           .30         ;Delay 96 us
            call            DELAYUS
            btfss           INPUT1      ;1 us
            goto            STARTTEST   ;2 us =>1+96+1+2 = 100 us
            return

;*************************************************************************************************
;Calibrate()-  Measures A Period From The Input(GP0) Reference Signal
;           -  Updates Osccal Value
;           -  Updates E^2
;*************************************************************************************************
CALIBRATE
            movlw           CALIBRATIONTIME
            movwf           COUNTER     ;Calibration Counter

LOW0
            btfsc           INPUT0      ;Wait To Sample Low Edge #0 (Makes Sure We Are
                                        ;Synchronized First Time Through)
            goto            LOW0

HIGH1
            btfss           INPUT0      ;Wait To Sample High Edge #1
            goto            HIGH1
            clrf            TMR0        ;Start Timer (Timer Will Be Behind By 5us After This
                                        ;Instruction)

LOW1
            btfsc           INPUT0      ;Wait To Sample Low Edge #1
            goto            LOW1

HIGH2
            btfss           INPUT0
            goto            HIGH2
            movf            TMR0,W      ;Stop Timer (Timer Will Be Stopped 3us Late)
            addlw           .2          ;Timer Is Behind By 2us Total From Start To Stop
            call            CHECKPERIOD ;See If Osccal Needs To Be Adjusted

LOW2
            btfsc           INPUT0      ;Wait To Sample Low Edge #2
            goto            LOW2
            decfsz          COUNTER,F   ;Decrement The Calibration Counter
            goto            HIGH1
            call            UPDATE_EE   ;Update E^2
            return

;*************************************************************************************************
;Update_EE - This routine Updates Calibration Flag & Calibration Value
;*************************************************************************************************
UPDATE_EE
            BANKSEL         EEDATA      ;BANK1
            movlw           0x5A        ;Update Calibration Flag
            movwf           EEDATA
            movlw           CALFLAGADR
            call            EEWRITE

            movf            OSCCAL,W
```

```
         movwf             EEDATA
         movlw             CALVALADR
         call              EEWRITE      ;Update Calibration Value
         BANKSEL           GPIO         ;BANK0
         return

;*****************************************************************************************
;CheckPeriod(W)- This routine computes the difference between the REFERENCEPERIOD and
;                MEASUREDPERIOD
;              - The MEASUREDPERIOD is contained in W when this routine is called
;              - The OSCCAL Is Adjusted Up or Down If The Period Is Outside The Specified
;                Tolerance
;*****************************************************************************************
CHECKPERIOD
         sublw             REFERENCEPERIOD
         btfsc             STATUS,Z     ;If (ReferencePeriod - MeasuredPeriod = 0) Don't Change
                                        ;OSCCAL
         return

         btfsc             STATUS,C     ;If (ReferencePeriod - MeasuredPeriod > 0) Oscillator
                                        ;Could Be Too Fast
         goto              RUNNINGSLOW  ;Else Oscillator Could Be Too Slow

RUNNINGFAST
         xorlw             0xFF         ;Two's Complement Value
         addlw             .1
         sublw             TOLERANCE    ;If (Tolerance - (ReferencePeriod - MeasuredPeriod) = 0
                                        ;Don't Change Osccal
         btfsc             STATUS,Z
         return
         goto              ADJUSTDOWN   ;Else Adjust Osccal Down

RUNNINGSLOW
         sublw             TOLERANCE    ;If (Tolerance - (ReferencePeriod - MeasuredPeriod) = 0
                                        ;Don't Change Osccal
         btfsc             STATUS,Z
         return
         goto              ADJUSTUP     ;Else Adjust Osccal Up

ADJUSTDOWN
         BANKSEL           OSCCAL       ;BANK1
         movlw             .4
         subwf             OSCCAL,F     ;Adjust Osccal Down
         BANKSEL           GPIO         ;BANK0
         return

ADJUSTUP
         BANKSEL           OSCCAL       ;BANK1
         movlw             .4
         addwf             OSCCAL,F     ;Adjust Osccal Up
         BANKSEL           GPIO         ;BANK0
         return
;*****************************************************************************************
;EEREAD(W) -  Address To Read Is Contained In W When This Function Is Called
;*****************************************************************************************
EEREAD
         BANKSEL           EEADR        ;BANK1
         movwf             EEADR
         bsf               EECON1,RD
         movf              EEDATA,W
         BANKSEL           GPIO         ;BANK0
         return

;*****************************************************************************************
;EEWRITE(W) -  Address To Read Is Contained In W When This Function Is Called
```

```
;           -  EEDATA Is Loaded Prior To This Function Call
;           -  BANK1 must be selected before this function is called
;*****************************************************************************************
EEWRITE
         movwf              EEADR
         bsf                EECON1,WREN
         bcf                INTCON, GIE
         movlw              0x55
         movwf              EECON2
         movlw              0xAA
         movwf              EECON2
         bsf                EECON1,WR
EECOMPLETE
         btfsc              EECON1,WR
         goto               EECOMPLETE
         bcf                EECON1,WREN
         return

;*****************************************************************************************
;*****************************************************************************************
;DELAYUS(W) -  Delay Microseconds
;           -  TotalTime(W) = [(1)+(2)+(1)+(W*3-1)+1+2] * (4/OSC) (This includes the movlw & the
;              call)
;           -  Max Time When W=0xFF, [ 771 Cycles * (4/OSC) ]
;           -  Must Declare COUNTER AS GPR
;           -  W > 0
;*****************************************************************************************
DELAYUS
         movwf              COUNTER
LOOP1
         decfsz             COUNTER,F
         goto               LOOP1
         nop
         return

;*****************************************************************************************
;EEPROM -  CALFLAGADR  - Contains Calibration Flag Value
;       -  CALVALADR - Contains Calibration Value
;*****************************************************************************************
         ORG                (0X2100+CALFLAGADR)

         DE                 0A5H            ;Initialize Calibration Flag

         DE                 000H            ;Initialize Calibration Value

         END
```

Note the following details of the code protection feature on Microchip devices:

- Microchip products meet the specification contained in their particular Microchip Data Sheet.
- Microchip believes that its family of products is one of the most secure families of its kind on the market today, when used in the intended manner and under normal conditions.
- There are dishonest and possibly illegal methods used to breach the code protection feature. All of these methods, to our knowledge, require using the Microchip products in a manner outside the operating specifications contained in Microchip's Data Sheets. Most likely, the person doing so is engaged in theft of intellectual property.
- Microchip is willing to work with the customer who is concerned about the integrity of their code.
- Neither Microchip nor any other semiconductor manufacturer can guarantee the security of their code. Code protection does not mean that we are guaranteeing the product as “unbreakable.”

Code protection is constantly evolving. We at Microchip are committed to continuously improving the code protection features of our products. Attempts to break microchip’s code protection feature may be a violation of the Digital Millennium Copyright Act. If such acts allow unauthorized access to your software or other copyrighted work, you may have a right to sue for relief under that Act.

DNV Certification, Inc. USA — ANSI-RAB QMS ACCREDITED — DNV — DNV MSC The Netherlands Accredited by the RvA — ISO 9001 / QS-9000 REGISTERED FIRM

Microchip received QS-9000 quality system certification for its worldwide headquarters, design and wafer fabrication facilities in Chandler and Tempe, Arizona in July 1999 and Mountain View, California in March 2002. The Company’s quality system processes and procedures are QS-9000 compliant for its PICmicro® 8-bit MCUs, KEELOQ® code hopping devices, Serial EEPROMs, microperipherals, non-volatile memory and analog products. In addition, Microchip’s quality system for the design and manufacture of development systems is ISO 9001 certified.

WORLDWIDE SALES AND SERVICE

AMERICAS

Corporate Office
2355 West Chandler Blvd.
Chandler, AZ 85224-6199
Tel: 480-792-7200 Fax: 480-792-7277
Technical Support: 480-792-7627
Web Address: http://www.microchip.com

Atlanta
3780 Mansell Road, Suite 130
Alpharetta, GA 30022
Tel: 770-640-0034 Fax: 770-640-0307

Boston
2 Lan Drive, Suite 120
Westford, MA 01886
Tel: 978-692-3848 Fax: 978-692-3821

Chicago
333 Pierce Road, Suite 180
Itasca, IL 60143
Tel: 630-285-0071 Fax: 630-285-0075

Dallas
4570 Westgrove Drive, Suite 160
Addison, TX 75001
Tel: 972-818-7423 Fax: 972-818-2924

Detroit
Tri-Atria Office Building
32255 Northwestern Highway, Suite 190
Farmington Hills, MI 48334
Tel: 248-538-2250 Fax: 248-538-2260

Kokomo
2767 S. Albright Road
Kokomo, Indiana 46902
Tel: 765-864-8360 Fax: 765-864-8387

Los Angeles
18201 Von Karman, Suite 1090
Irvine, CA 92612
Tel: 949-263-1888 Fax: 949-263-1338

Phoenix
2355 West Chandler Blvd.
Chandler, AZ 85224-6199
Tel: 480-792-7966 Fax: 480-792-4338

San Jose
Microchip Technology Inc.
2107 North First Street, Suite 590
San Jose, CA 95131
Tel: 408-436-7950 Fax: 408-436-7955

Toronto
6285 Northam Drive, Suite 108
Mississauga, Ontario L4V 1X5, Canada
Tel: 905-673-0699 Fax: 905-673-6509

ASIA/PACIFIC

Australia
Microchip Technology Australia Pty Ltd
Marketing Support Division
Suite 22, 41 Rawson Street
Epping 2121, NSW
Australia
Tel: 61-2-9868-6733 Fax: 61-2-9868-6755

China - Beijing
Microchip Technology Consulting (Shanghai) Co., Ltd., Beijing Liaison Office
Unit 915
Bei Hai Wan Tai Bldg.
No. 6 Chaoyangmen Beidajie
Beijing, 100027, No. China
Tel: 86-10-85282100 Fax: 86-10-85282104

China - Chengdu
Microchip Technology Consulting (Shanghai) Co., Ltd., Chengdu Liaison Office
Rm. 2401-2402, 24th Floor,
Ming Xing Financial Tower
No. 88 TIDU Street
Chengdu 610016, China
Tel: 86-28-86766200 Fax: 86-28-86766599

China - Fuzhou
Microchip Technology Consulting (Shanghai) Co., Ltd., Fuzhou Liaison Office
Unit 28F, World Trade Plaza
No. 71 Wusi Road
Fuzhou 350001, China
Tel: 86-591-7503506 Fax: 86-591-7503521

China - Hong Kong SAR
Microchip Technology Hongkong Ltd.
Unit 901-6, Tower 2, Metroplaza
223 Hing Fong Road
Kwai Fong, N.T., Hong Kong
Tel: 852-2401-1200 Fax: 852-2401-3431

China - Shanghai
Microchip Technology Consulting (Shanghai) Co., Ltd.
Room 701, Bldg. B
Far East International Plaza
No. 317 Xian Xia Road
Shanghai, 200051
Tel: 86-21-6275-5700 Fax: 86-21-6275-5060

China - Shenzhen
Microchip Technology Consulting (Shanghai) Co., Ltd., Shenzhen Liaison Office
Rm. 1812, 18/F, Building A, United Plaza
No. 5022 Binhe Road, Futian District
Shenzhen 518033, China
Tel: 86-755-82901380 Fax: 86-755-82966626

China - Qingdao
Rm. B505A, Fullhope Plaza,
No. 12 Hong Kong Central Rd.
Qingdao 266071, China
Tel: 86-532-5027355 Fax: 86-532-5027205

India
Microchip Technology Inc.
India Liaison Office
Marketing Support Division
Divyasree Chambers
1 Floor, Wing A (A3/A4)
No. 11, O'Shaugnessey Road
Bangalore, 560 025, India
Tel: 91-80-2290061 Fax: 91-80-2290062

Japan
Microchip Technology Japan K.K.
Benex S-1 6F
3-18-20, Shinyokohama
Kohoku-Ku, Yokohama-shi
Kanagawa, 222-0033, Japan
Tel: 81-45-471- 6166 Fax: 81-45-471-6122

Korea
Microchip Technology Korea
168-1, Youngbo Bldg. 3 Floor
Samsung-Dong, Kangnam-Ku
Seoul, Korea 135-882
Tel: 82-2-554-7200 Fax: 82-2-558-5934

Singapore
Microchip Technology Singapore Pte Ltd.
200 Middle Road
#07-02 Prime Centre
Singapore, 188980
Tel: 65-6334-8870 Fax: 65-6334-8850

Taiwan
Microchip Technology (Barbados) Inc., Taiwan Branch
11F-3, No. 207
Tung Hua North Road
Taipei, 105, Taiwan
Tel: 886-2-2717-7175 Fax: 886-2-2545-0139

EUROPE

Austria
Microchip Technology Austria GmbH
Durisolstrasse 2
A-4600 Wels
Austria
Tel: 43-7242-2244-399
Fax: 43-7242-2244-393

Denmark
Microchip Technology Nordic ApS
Regus Business Centre
Lautrup hoj 1-3
Ballerup DK-2750 Denmark
Tel: 45 4420 9895 Fax: 45 4420 9910

France
Microchip Technology SARL
Parc d'Activite du Moulin de Massy
43 Rue du Saule Trapu
Batiment A - ler Etage
91300 Massy, France
Tel: 33-1-69-53-63-20 Fax: 33-1-69-30-90-79

Germany
Microchip Technology GmbH
Steinheilstrasse 10
D-85737 Ismaning, Germany
Tel: 49-89-627-144-0
Fax: 49-89-627-144-44

Italy
Microchip Technology SRL
Via Quasimodo, 12
20025 Legnano (MI)
Milan, Italy
Tel: 39-0331-742611 Fax: 39-0331-466781

United Kingdom
Microchip Ltd.
505 Eskdale Road
Winnersh Triangle
Wokingham
Berkshire, England RG41 5TU
Tel: 44 118 921 5869 Fax: 44-118 921-5820

03/25/03